
Clear Globals(); 
 
dlg = Column Dialog( 
 yCol = Col List( "Variable", 
  Data Type( Numeric ), 
  Min Col(1), 
  Max Col(2) 
 ), 
  
 Line Up( 2, 
  "Number of Samples", k = Edit Number( 2000 ), 
  "Quantile to Bootstrap",q = Edit Number(.10) 
 ), 
 
 V List( 
  "Demonstration", 
  demo = Radio Buttons( "Test Single Quantile", "Compare Two 
Quantiles" ) 
 ), 
 
 "", "Select columns for resampling demonstration" 
); 
 
 
If( dlg["Button"] == -1, Throw( "User cancelled" ) ); 
Remove From( dlg ); Eval List( dlg ); 
 
If( demo == 2 & N Items( yCol ) != 2, 
 Dialog( 
  "Only one column provided to compare two quantiles", "", 
  Button( "OK" ) 
 ); 
 Throw( "Only one column provided to compare two quantiles" ); 
); 
 
dt = Current Data Table(); 
nn = N Row(); 
 
// create containers for resampling results. 
m = J( k, 1, . ); 
 
// get 'population'. 
y = dt << Get As Matrix( yCol ); 
 
// collect sample statistics. 
Choose( demo, 
 // test one quantile. 
 y1 = y[0,1];  // just use the first column. 
 For( s = 1, s <= k, s++, 
  m[s] = Quantile(q, y1[ J( nn, 1, Random Integer( nn ) ) ] ); 
 ), 
 // compare two quantiles. 
 y1 = y[0,1]; 
 y2 = y[0,2]; 
 For( s = 1, s <= k, s++, 
  m[s] = Quantile(q, y2[ J( nn, 1, Random Integer( nn ) ) ] ) - 
   Quantile(q, y1[ J( nn, 1, Random Integer( nn ) ) ] ); 



 ); 
); 
 
// save and show the statistics. 
rdt = New Table( "Resample Results", 
 New Column( "Sample", Values( 1::k ) ), 
 New Column( "Quantile", Values( m ) ), 
); 
 
 
If( demo == 2, 
 Column( 2 ) << Set Name( "Difference" ); 
); 
 
dist = rdt << Distribution( Y( Column( 2 ) ), 
Moments(0) ); 
dist << Set Title( "Bootstrap Statistics" ); 
 


